

パソコンを用いた比色計による環境調査の教材化Ⅱ

— 小学校における酸性雨測定の授業実践 —

鳥取大学教育学部理科教育 杉 本 良 一
大阪府教育センター 紺 野 昇*

Study on the Teaching Material for Air Pollution Investigation with Computer II

— Practical Teaching of Acid Rain Measurement for Elementary School Science —

Ryoichi SUGIMOTO, Noboru KONNO*

1. はじめに

ここ数年、環境教育の重要性が指摘され、積極的な推進が図られている。理科における環境教育は、観察及び実験を行い、自然についての理解を深め、探究的な能力の育成と科学的な見方や考え方の養成という理科教育の目標を基盤として行われる必要がある^{1),2)}。

環境調査は、理科教育における大切な学習活動であるが、一般に小・中学校の授業の中で、汚染物質の濃度を定量的に分析することは困難であると考えられており、滴定法や比色法を用いて、定量的に環境を調べるような学習活動はあまり行われてこなかった³⁾。

汚染の有無だけを調べるような、定性的な扱いだけを学習活動の中で行くと、児童・生徒に環境に対する恐れや憂慮など、否定的感情だけを起こしかねないことが考えられる。

環境教育の視点で行われる環境調査の授業実践は、科学的かつ定量的な扱いによって、自分たちの身のまわりの環境について、総合的に判断を行い、児童・生徒の環境への意識を大きく高めることができると思う。

我々は前報⁴⁾で、汚染物質を簡単な操作で、定量的に測定する手段として、自作の比色計とコンピュータを用いた環境調査の教材システムを報告した。この比色計(写真1)は、500mlポリ製試薬瓶とフィルムケースを本体として、発光ダイオードと光センサーで製作したもので、大気中の NO_x 及び水質中の亜硝酸イオン・リン酸イオンの測定を行うことができる。

現在まで、大阪府下を中心とした多数の教員がこの比色計を製作しており、主に小・中学校の理

科授業で実際に利用されている。その中で、本システムを用いて、身のまわりにある他の汚染物質の測定もできないかという要請があり、比色計を用いて酸性雨等のpH測定を行うシステムを検討し、開発した。

現在、車の排気ガスや工場の排煙による NO_x ・ SO_x の発生が増えた結果、大気汚染が起こり、酸性雨（pH5.6以下の雨）の問題が生じている。この酸性雨の問題については、小学校5年の社会科や、中学校の理科（第2分野下）でも教材として取り上げられている。このような身近に起こっている環境汚染の測定は、本比色法によって簡単に行えるので、小学校から高等学校までの環境調査の教材として利用が期待できる。

今回、小学校において、本システムを使用した酸性雨を測定する授業を行った。この授業を通して、子どもたちの酸性雨に対する理解内容の変化や、雨に対する情意の変化、コンピュータの利用に対する意識を調査したので併せて報告する。




写真1 本計測システム

2. 酸性雨のpH測定

(1) 本システムのねらい

一般に雨水の酸性度を測定するには、pHメーターを用いることが多いが、操作方法は煩雑で子どもたちが使用することは困難であることが多い。また、pHメーターは高価であり、小・中学校での経済的な負担は大きい。最近の簡易型のpHメーターは安価であるが、測定誤差が0.1~0.2と大きく、酸性雨のpH測定には十分とは言えない。

そこで、最近学校に積極的に導入されているコンピュータを利用して、コンピュータと比色計を用いた定量的なpH測定の方法を開発した。比色測定に利用できる酸・塩基指示薬（以降、指示薬と表示する）を調べたところ、いくつかの指示薬によりpHの定量的な測定が可能であることが分かった。

pHメーターを用いた測定では、操作方法は機械的で、子どもたちの感動は期待しにくい。本法では指示薬による試料水の色の変化が顕著であり、コンピュータを用いて精度よく測定できるなど、子どもたちの興味・関心を高める効果も期待できる。

(2) 測定原理

比色法によるpH測定は、一定量の試料水に一定の指示薬を加えて発色させ、550nmの波長の吸光度を比色計で測定する。特定の指示薬は、一定波長の吸光度とpHとの間に相関関係があり、吸光度からコンピュータの解析処理によってpHが換算できる。ハードウェア及びソフトウェアの詳細は前報⁴⁾及び文献⁵⁾⁻⁸⁾を参照されたい。

分光光度計(島津UV240)によりBCP(プロムクレゾールパープル)指示薬について、各pHにおける可視光の吸収スペクトルを調べた結果を図1に示す。図2は、この吸収スペクトルのうち、本比色計で使用する緑色発光ダイオードの波長(550nm)の部分について、吸光度とpHとの関係を調べたものである。この波長の吸光度とpHには指示薬の変色領域内で相関が認められ、吸光度からpH値の算出が可能であることが分かる。




図1 BCPの各pH溶液の吸収スペクトル




図2 550nmの波長の吸光度とpHの関係

その他の指示薬について同様に調べた結果、550nmの波長の光によってpHと相関のある吸光を行うことができる指示薬を表1に示す。これらの指示薬を用いると、pH3～9の範囲で精度よく測定できる。

表1 比色法で用いる指示薬と測定範囲

指 示 薬	変色範囲*	色 の 変 化
MO (メチルオレンジ)	3.1～4.4	黄色～赤色
MR (メチルレッド)	4.4～6.2	赤色～黄色
BCP (ブムロクレゾールパープル)	5.2～6.8	黄色～紫色
CR (クレゾールレッド)	7.2～8.8	黄色～赤色

*単位：pH

図3は、本比色計でpHを測定するために用いるBCPの検量線である。縦軸が本比色計で測定したコンピュータの吸光度に関連するカウントデータで、横軸がpHである。図より、良好な直線関係が得られることが分かった。また、様々なpHの溶液について、本比色計の測定とpHメーター（東亜、HM-7E）で測定した結果を比較したものを図4に示す。2つの図から、本比色計によるpH測定は、十分に利用できることが分かる。

なお、使用する指示薬の濃度は、試料水10mlに対して0.1%の指示薬0.2ml加えたものである。




図3 本比色計でpH算出のためのBCPの検量線




図4 本比色計とpHメーターによるpH測定の比較

(3) 検量線データファイルの作成方法

本比色計の使用には、使用するコンピュータの処理能力を登録した検量線のデータファイルを、次の手順に従って指示薬別に作成する必要がある。

① まず検量線を作成する指示薬について、変色範囲の上限と下限のpH値のリン酸標準液を準備する。

BCPでは5.2と6.8, MRでは4.4と6.2, MOでは3.1と4.4, CRでは7.2と8.8である。

② 各標準液10mlを試験管にとり、指示薬を0.2ml加える。

- ③ コンピュータを起動し、ディスクを立ちあげる。

指示薬の選択画面で、検量線を作成する指示薬を選び、処理画面で検量線作成を選択する。最初に、強酸側溶液の試験管を比色計にセットし、画面の指示に従ってそのpH値をキー入力する。次に、弱酸側の試験管を比色計にセットし、そのpH値をキー入力する。

以上の手順を、各指示薬について同様に行う。なお、資料1にBASICプログラムの一部を示す。

(4) 測定方法

- ① 試料水10mlを試験管に取り、BCP指示薬を0.2ml加える。
- ② コンピュータを起動し、指示薬の選択画面でBCPを選択する。比色計に試料水の試験管を入れ、画面の指示に従って「測定」を選ぶ。測定したpH値が画面に表示され、STOPキーを押すと測定を終了する。
- ③ BCP指示薬では、pHが5.2以上の強酸性の場合（溶液は黄色）には測定値は頭打ちとなって、それ以上の強酸性（小さい値）を示さない。
- そこで、BCP指示薬を加えて液が黄色になり、pHが5.2に近い場合は、より強酸の可能性があるため、MR（メチルレッド）指示薬を用いて再度測定する必要がある。MR指示薬を用いて同様に測定し、BCP指示薬より小さい値の場合は、その値が試料水のpHである。
- ④ また、MR指示薬を用いてpHが4.4に近い場合（溶液は赤色）は、同様に測定値が頭打ちとなっていることが考えられる。その場合は、MO（メチルオレンジ）指示薬に変えて実験を行う。
- 最近の雨水は、年間を通じて4.4以下になることはあまりないので、BCPとMRだけで十分に測定できる。

(5) 測定結果

平成7年5月より11月までに降った5回の雨水を、大阪府教育センター屋上で採取し、本システムで測定した結果、平均値は5.2、最低値が3月20日の4.4、最高値が5月24日の5.9であった。

3. 小学校での本システムを用いた酸性雨測定の授業実践

平成7年10月から11月にかけて、大阪狭山市立第七小学校において、本比色計を用いた授業を実施した。

(1) 指導目標

小学校5年理科の単元「天気の変化」において、「雲と雨」についての発展授業として、酸性雨の測定を行った。指導目標は、次のとおりである。

- ① 私たちのまわりに降る雨に、酸性雨が検出されることを、実際の測定を通して確かめる。
- ② 車の排気ガスや工場の排煙による大気汚染の結果、酸性雨の現象が起きていることを理解する。

(2) 指導計画

- ① 導入部：アサガオの花びらについての赤い斑点の原因を考えた後、雨を調べてみることにする。雨の日に子どもたちが雨を集める。なお、事前に酸性度の目安となるpHについての学習を行う。
- ② 展開部：集めた雨水を、本比色計によって子どもたち自身がpHの測定を行う。
- ③ 結論部：測定した結果を集約し、自分たちのまわりに酸性雨が降ったかどうかを考えさせる。

また、酸性度の分布を調べ、その原因を考察する。

(3) 測定結果

1995年(平成7年)10月24日の測定では、表2のとおり5.1~5.4までの酸性雨を観測した。測定データは、国道沿いの場所と、そうでない場所に分類して集計した。

平均値から、国道沿いの場所の方がpHが小さく、より酸性になる傾向が見られた。

表2 大阪狭山市立小学校で測定した酸性雨のpH結果

測定場所	測定値*	平均値
国道沿いの測定点 (5箇所)	5.1, 5.1, 5.2, 5.2, 5.2	5.2
国道沿いでない測定点 (8箇所)	5.1, 5.2, 5.3, 5.3, 5.3, 5.3, 5.4, 5.4	5.3

*単位：pH

(4) 酸性雨に対する授業後の理解

授業の前後に、酸性雨の原因についてのアンケート調査を行ない、環境調査後に学習の理解が進むかどうかを調べた。酸性雨の原因についての質問に、3つの解答例を示し、それぞれを原因と思うかどうかを「はい」と「いいえ」で答える方法で実施した。回答の重複を認めており、表3がその結果である。

授業後、ゴミの多さや自然の減少が原因と回答した割合は減少し、全員が車の排気ガスや工場の煙が原因と回答することができた。

表3 酸性雨の原因に関する質問への「はい」の回答率(%)

(対象35人)

酸性雨の原因についての質問内容	授業前	授業後
酸性雨の原因はゴミが多すぎるため	40.6%	12.5%
酸性雨の原因は自然が減ったため	40.6%	21.9%
酸性雨の原因は車の排気ガス、工場の煙	96.9%	100%

(5) 調査後における児童の情意面の変化

酸性雨を測定する授業を通して、子どもたちの雨に対する意識がどのように変化するか、情意面の変化について調査した。調査は、授業前後に表4の項目についてアンケート調査を行った。

授業後、「雨は恐いですか」という質問に対する回答率の変化はなかったが、「雨は汚いと思いますか」という質問に「はい」と回答した割合は、76%から91%に増加した。また、「雨が酸性になると、飲み水も酸性になると思いますか」に対する「はい」の回答は、25%から16%に減少した。

この結果、今回の酸性雨を測定する体験を通して、雨の汚れを認識する子どもの割合は増加したことが分かった。しかし、雨の汚れを感じても、雨が怖いというような否定的感情を持つ子どもの割合は増えなかった。

表4 雨に対する意識・情意の調査結果

(対象35人)

質 問 項 目	授 業 前	授 業 後
雨が汚いと思いますか	75.8%	90.6%
雨が怖いと思いますか	62.5%	62.5%
雨が酸性になると、飲み水も酸性になると 思いますか	25.0%	12.5%

(6) コンピュータを用いた実験に対する子どもたちの意識

コンピュータを用いた実験に対する子どもたちの意識について、授業後図5のアンケート調査を行った。コンピュータを用いた授業が、「楽しい」と回答した割合は63%で、「少し楽しい」を合わせた割合は90%を越えた。「他の学習にコンピュータを利用したいか」という質問には80%以上が「利用したい」と回答し、「少し利用したい」の割合を合わせると100%になった。この結果、コンピュータを用いた実験に対して、子どもたちは好意的であり、また意欲的であることが分かった。

次に、「コンピュータの測定内容は理解できたか」という質問に、「理解できた」と「少しできた」と回答した合計は97%であった。この結果、本教材システムは、小学校においても十分に利用できると思われる。




図5 コンピュータを用いた実験での子どもたちの意識調査結果

5. おわりに

今回行った雨の酸性度を定量的に測定する学習活動では、子どもたちに「雨が怖い」という酸性雨に対する否定的な感情の増加は見られず、事前に行ったpHの学習を通じて、どの程度の酸性であるかという理解がなされたと考える。

従来、教育活動で行う科学的な環境調査は、パックテストを中心とした半定量的なものが中心であるが、コンピュータの解析処理を用いて行う比色分析法は測定精度が高く、アンケート調査の結果から判断して、操作方法と測定結果に対する理解は子どもたちにとっては難しくなかった。また、コンピュータを用いた授業に対する子どもたちの期待は、たいへん高いことも分かった。

本比色計によって、現在までに大気中の NO_x 、水質中の $\text{NO}_3^- \cdot \text{PO}_4^{3-}$ 、雨水などのpH測定が可能となったが、今後は光化学スモッグの原因となるオキシダントや、水質中のCODの測定方法を開発する予定である。本教材システムを用いて、環境教育の視点に立った環境調査が多数実践されることを期待している。

最後に本研究にあたり、実践協力して頂いた大阪狭山市立第七小学校の大塚淳子教諭に感謝いたします。

引用・参考文献

- 1) 文部省：環境教育指導資料（小学校編），大蔵省印刷局，1993
- 2) 谷 友雄：小学校における理科を通しての環境教育，理科の教育，第42巻(7)，1993
- 3) 総合教育技術増刊：環境教育ガイド'95～'96，小学館，p4，1995
- 4) 杉本良一，紺野昇，鳥海重治：鳥取大学教育学部研究報告，第37巻，第1号，pp.1-20，1995
- 5) 紺野 昇：「パソコンを用いた自作比色計による環境調査」，化学と教育，日本化学会，第43巻8，pp.527-530，1995
- 6) 紺野 昇，大塚淳子，杉本良一：日本理科教育学会研究紀要，第36巻(2)，pp.11-20，1995
- 7) 杉本良一：「マウス端子を用いた簡単な化学計測用インターフェース」，化学と教育，第41巻8号，pp.558-559，1993
- 8) 紺野 昇：「A/D変換器を用いない安価なパソコン計測の応用」，化学と教育，第43巻9号，p.600，1995

(1996年4月19日受理)

英文要約

SUMMARY

Study on the Teaching Material for Air Pollution Investigation with Computer II —Practical Teaching of Acid Rain Measurement for Primary School Science—

Faculty of Education, Tottori University
Ryoichi SUGIMOTO
Osaka Prefectural Education Center
Noboru KONNO

An environmental investigation is an important learning activity in the science education.

Air pollution happens since NO_x • SO_x with the exhaust gas of the car and the smoke of the factory increases and the problem of acid rain is caused now.

Social studies of 5-th grade in the primary school and the science of the lower secondary school are taken up as a teaching material about the problem of the acid rain.

In general, it is difficult to analyze the density of the contaminant quantitatively in the class of the primary and the lower secondary school.

This activity to examine the environment quantitatively by using the titration method and colorimeter method had not been done so much.

We reported on hand-made colorimeter and the teaching material system of an environmental investigation with a computer with the previous thesis as a means to measure the contaminant quantitatively. This time, the system of acid rain which measured pH was developed with this measurement method.

Because the environmental pollution which happens near oneself can be easily measured by this colorimeter method, the use can be expected as a teaching material of an environmental investigation from the primary school to the upper secondary school.

In addition, practical teaching used this system was also conducted in the primary science class. As a result, the affective aspect of the pupils understanding to acid rain and consideration to the use of the computer were improved.

```

10 'save "PH1"
20 '*****
30 '      比色濃度計測 PC-9801マウスインターフェース使用型
40 '      酸性度測定システム（弱酸性タイプ：BCP）
50 '      大阪府教育センター 紺野 昇      1995.9.18
60 '*****
70 DEF SEG = &H8000
80 SCREEN 3,0:COLOR 7,0,,2:CONSOLE 0,25,0,1:WIDTH 80,25
90 '
100 CLS 3:DIM X(1000),W(100,2),WD$(100)
110 '
120 COLOR=(8,&HA)
130 COLOR=(9,&H9A)
140 DR$=""      'データ記録ドライブ
150 FIL$=DR$+"D-PH.DAT" 'データ記録ファイル
160 FI$=DR$+"PH1.DAT"   '校正データ記録ファイル
170 TM=1      '測定時間間隔
180 CO=6      'グラフ表示色
190 DR$=""
200 '
210 ON STOP GOSUB *DERU:STOP OFF
220 GOSUB *FILEINP
230 GOTO *SHORI
240 '
250 *OWARI
260 RUN "PHMENU"
270 '
280 *WAK      '表示画面
290 TI$="コンピュータによる酸性度の測定（指示薬：BCP）"
300 TIME1$="":DATE1$=""
310 CLS 3
320 LINE(0,0)-(639,348),8,BF
330 LINE(0,354)-(639,399),9,BF
340 WY1=40:WY2=315:WX1=52:WX2=262      '図の左上 Y軸位置
350 LINE(WX1,WY1)-(WX2,WY2),0,BF
360 LINE(WX1,WY1)-(WX2,WY2),7,B
370 LINE(WX1-1,WY1-1)-(WX2+1,WY2+1),7,B
380 '
390 FOR I=1 TO 11
400 LINE(WX1,WY1+25*I)-STEP(WX2-WX1,0),5,,&H8888

```

```
410 NEXT I
420 FOR I=WY2 TO WY1 STEP -5
430 LINE(WX1,I)-STEP(5,0),7
440 NEXT I
450 FOR I=WX1 TO WX2 STEP 30
460 LINE(I,WY1)-(I,WY2),5,,&H4040
470 NEXT I
480 '
490 WX3=274:WX4=380
500 LINE(WX3,WY1)-(WX4,WY2+10),0,BF
510 LINE(WX3,WY1)-(WX4,WY2+10),7,B
520 LINE(WX3-1,WY1-1)-(WX4+1,WY2+11),7,B
530 LINE(WX3,WY1+28)-(WX4,WY1+28),7:PAINT(WX3+20,WY1+10),9,7
540 '
550 WX5=392:WX6=612
560 LINE(WX5,WY1)-(WX6,WY2+10),0,BF
570 LINE(WX5,WY1)-(WX6,WY2+10),7,B
580 LINE(WX5-1,WY1-1)-(WX6+1,WY2+11),7,B
590 LINE(WX5,WY1+28)-(WX6,WY1+28),7:PAINT(WX5+20,WY1+10),9,7
600 '
610 '
620 GOSUB *HOUJI
630 GOSUB *YOKO
640 RETURN
650 '
660 '
670 *TATE
680 CC1=7:CC2=5:TT=-.2          '縦軸目盛りの上下限
690 LINE(4,WY1-6)-(WX1-1,WY2+4),8,BF
700 ICH=WY2-5
710 FOR I=1 TO 11
720 Q=CC1+(I-1)*TT
730 L1=ICH:L0=8
740 TITL$=STR$(Q)
750 IF Q=7 THEN TITL$="7.0"
760 IF Q=6 THEN TITL$="6.0"
770 IF Q=5 THEN TITL$="5.0"
780 GOSUB *TITL
790 ICH=ICH-25
800 NEXT I
```

```
810 WW1=250/(CC1-CC2)
820 WW2=65-CC2*WW1
830 RETURN
840 '
850 *YOKO
860 LINE(WX1-2,WY2+2)-(WX2,WY2+18),8,BF
870 FOR L=WX1 TO WX2-30 STEP 30
880 L1=WY2+8:IF L=WX1 THEN L0=L-20 ELSE L0=L-24
890 LL=(L-WX1)/2
900 TITL$=STR$(LL)
910 GOSUB *TITL
920 NEXT L
930 RETURN
940 '
950 *HOUI
960 CLS
970 COLOR 6:LOCATE 5,1:PRINT TI$
980 COLOR 7:LOCATE 49,1:PRINT "日付"+DATE$;
990     LOCATE 64,1:PRINT "時刻"+TIME$
1000 '
1010 LOCATE 30,20:PRINT "(秒)":LOCATE 2,2:PRINT "(pH)";
1020 LOCATE 36,3:PRINT "測定データ"
1030 COLOR 7:LOCATE 35,6:PRINT "[酸性度]"
1040 COLOR 6:LOCATE 35,8:PRINT "pH = 0.00"
1050 COLOR 7:LOCATE 36,11:PRINT "[カウント数]"
1060 COLOR 6:LOCATE 37,13:PRINT "  0"
1070 COLOR 7:LOCATE 35,16:PRINT "[計測時間]"
1080 COLOR 6:LOCATE 37,18:PRINT "  0秒"
1090 '
1100 COLOR 7:LOCATE 56,3:PRINT "記録データ"
1110 RETURN
1120 '
1130 *DATOUT
1140 FOR I=1 TO 15
1150 IF DTNO=<15 THEN II=I ELSE II=DTNO-15+I
1160 LOCATE 49,4+I:COLOR 5:PRINT USING "###:";W(II,1);
1170 COLOR 7:PRINT "酸性pH=";
1180 COLOR 5:PRINT USING "#.##:";W(II,2);
1190 COLOR 7:PRINT WD$(II);
1200     IF DTNO=II THEN GOTO 1220
```

```

1210 NEXT I
1220 RETURN
1230 '
1240 *SHOKI
1250 OPEN FI$ FOR INPUT AS #1
1260 INPUT #1, A, B, C1, N1, C2, N2
1270 CLOSE #1
1280 RETURN
1290 '
1300 *SHORI ' //////////////////////////////////////
1310 GOSUB *WAK
1320 GOSUB *TATE
1330 LOCATE 1,23:PRINT SPC(78);
1340 LOCATE 10,23:COLOR 5
1350 PRINT "<<選択>>";
1360 COLOR 5:PRINT " 1:":COLOR 7:PRINT "濃度測定";
1370 COLOR 5:PRINT " 2:":COLOR 7:PRINT "検量線作成";
1380 COLOR 5:PRINT " 0:":COLOR 7:PRINT "終了";
1390 AS$=INKEY$:IF AS$="" THEN 1390
1400 LOCATE 1,23:PRINT SPC(77);
1410 IF AS$=CHR$(13) OR AS$="1" THEN OPW=1:GOTO *DATINP
1420 IF AS$="2" THEN OPW=3:GOSUB *MAKE
1430 IF AS$="0" THEN GOTO *OWARI ELSE GOTO 1390
1440 '
1450 *DATINP' 測定 //////////////////////////////////////
1460 LOCATE 1,23:PRINT SPC(77);:LOCATE 6,23:COLOR 7
1470 PRINT "<<測定:試料を比色計に入れ、準備ができたなら":COLOR 5:PRINT "リターン
キー";
1480 COLOR 7:PRINT "で測定します。>>";
1490 AS$=INKEY$:IF AS$="" THEN 1490
1500 LOCATE 1,23:PRINT SPC(76);
1510 '
1520 ON ERROR GOTO *EE
1530 GOSUB *SHOKI
1540 ON ERROR GOTO 0
1550 CNT=0:CTT=0
1560 STOP ON
1570 LOCATE 1,23:PRINT SPC(70);
1580 COLOR 7:LOCATE 20,23:PRINT "<<ただ今測定中。(":COLOR 5:PRINT "STOPキ
ー":COLOR 7:PRINT "で計測を中止)>>";

```

```

1590 '
1600 TM1=VAL(RIGHT$(TIME$,2))
1610 COLOR 7:LOCATE 64,1:PRINT "時刻 "+TIME$
1620 GOSUB *MOUSEIN
1630 TM2=VAL(RIGHT$(TIME$,2))
1640 IF TM2=TM1 THEN 1620
1650 IF TM2<TM1 THEN TM2=TM2+60
1660 TTT=TM2-TM1
1670 DAT=(LOG(COU)-A)/B          '濃度計算
1680 IF DAT<0 THEN DAT=0
1690 BEEP 1
1700 CNT=CNT+1:IF CNT>1000 THEN GOTO *DERU
1710 CTT=CTT+TTT
1720 WWI=DAT
1730 '
1740 YYT=WW1*WWI+WW2:IF YYT>WY2 THEN YYT=WY2
1750 IF CTT=<105 THEN CIRCLE(CTT*2+WX1-1,YYT),1,CO,,,F:GOTO 1800
1760 IF CTT>105 AND CNT<210 THEN CIRCLE(CTT*2+WX1-208,YYT),1,CO-2,,,F:
GOTO 1800
1770 WT=CTT MOD 105:IF WT=0 THEN WT=105
1780 CIRCLE(WT*2+WX1-1,YYT),1,CO-4,,,F
1790 '
1800 COLOR 6
1810 LOCATE 41,8:PRINT USING "#.##":DAT;
1820 LOCATE 38,13:PRINT USING "####";COU
1830 LOCATE 38,18:PRINT USING "###";CTT
1840 BEEP 0
1850 IF OPW=2 THEN GOTO *SHORI
1860 GOTO 1600
1870 '
1880 *DERU
1890 STOP OFF
1900 IF OPW=2 THEN GOTO *SHORI
1910 LOCATE 1,23:PRINT SPC(70);
1920 COLOR 5:LOCATE 6,23:PRINT "<< 選択 >> ";
1930 COLOR 7:PRINT " 1:測定データを記録  2:測定を終了  リターンキー:測定継続";
1940 COLOR@(19,23)-(21,23),5
1950 COLOR@(41,23)-(43,23),5
1960 COLOR@(57,23)-(63,23),5
1970 AS$=INKEY$:IF AS$="" THEN GOTO 1970

```



```
1980 IF AS$="2" THEN GOTO *SHORI
1990 IF AS$=CHR$(13) THEN GOSUB *KESU:RETURN 1550
2000 IF AS$="1" THEN GOTO *KIROKU ELSE GOTO 1970
2010 '
2020 *KESU
2030 LINE(WX1,WY1)-(WX2,WY2),7,B
2040 PAINT(WX1+30,WY1+100),0,7
2050 FOR I=1 TO 11
2060 LINE(WX1,WY1+25*I)-STEP(WX2-WX1,0),5,,&H8888
2070 NEXT I
2080 FOR I=WX1 TO WX2 STEP 30
2090 LINE(I,WY1)-(I,WY2),5,,&H4040
2100 NEXT I
2110 RETURN
2120 '
2130 *KIROKU
2140 LOCATE 1,23:PRINT SPC(77);
2150 COLOR 7:LOCATE 8,23:PRINT "* 記録する";:COLOR 5:PRINT " 試料の番号 ";:COL
OR 7:PRINT "を入力して下さい。(0 で中止) => ";
2160 INPUT WNN
2170 IF WNN=0 THEN GOTO 1880
2180 LOCATE 1,23:PRINT SPC(77);
2190 COLOR 7:LOCATE 24,23:PRINT "* ただ今、データを記録中です。*";
2200 IF DTNO=0 THEN 2270
2210 FOR I=1 TO DTNO
2220 IF W(I,1) <> WNN THEN 2260
2230 GOSUB *MES2
2240 IF OPE=1 THEN GOTO 2270
2250 IF OPE=2 THEN GOTO 2140
2260 NEXT I
2270 DTNO=DTNO+1:W(DTNO,1)=WNN:W(DTNO,2)=WWI:WD$(DTNO)=DATE$
2280 GOSUB *DATOUT
2290 ON ERROR GOTO *ER1
2300 OPEN FIL$ FOR INPUT AS #1:INPUT #1,DUM:CLOSE
2310 ON ERROR GOTO 0
2320 OPEN FIL$ FOR APPEND AS #1
2330 WRITE #1,WNN,WWI,DATE$
2340 CLOSE #1
2350 GOTO 1880
2360 '

```

```

2370 *ER1
2380 CLOSE
2390 OPEN FIL$ FOR OUTPUT AS #1
2400 RESUME 2330
2410 '
2420 *MES2
2430 LOCATE 2,23:PRINT SPC(77);:COLOR 6
2440 ON ERROR GOTO 0
2450 LOCATE 6,23:PRINT "<< 指定した番号は、既に登録されています。 1:上書き 2:番
号変更 >>";
2460 AS$=INKEY$:IF AS$="1" THEN OPE=1:GOTO 2480
2470 IF AS$="2" THEN OPE=2 ELSE 2460
2480 RETURN
2490 '
2500 *FILEINP'////////////////////////////////////
2510 ON ERROR GOTO *EE
2520 GOSUB *SHOKI
2530 ON ERROR GOTO 0
2540 RETURN
2550 '
2560 *TITL
2570 FOR AV=1 TO LEN(TITL$)
2580 GPO=ASC(MID$(TITL$,AV,1))
2590 PUT@(L0+AV*8,L1),KANJI(GPO+&H100),,7,0
2600 NEXT AV
2610 RETURN
2620 '
2630 *EE
2640 STOP OFF
2650 CLOSE:CLS 3:COLOR 6
2660 LOCATE 18,6:PRINT "警告！ 検量線ファイルが記録されていません。"
2670 COLOR 7:LOCATE 20,11:PRINT "検量線ファイルを作成してから測定して下さい。"
2680 LOCATE 20,16:PRINT "* 確認したら リターンキー を押して下さい。*";
2690 AS$=INKEY$:IF AS$="" THEN GOTO 2690
2700 GOTO *SHORI
2710 '
2720 *MAKE
2730 '*****
2740 ' マウス端子使用 比色濃度計測システム 較正プログラム
2760 '*****

```

```

2770 CLS 3
2780 LINE(20,0)-(620,16),1,BF:LOCATE 11,0
2790 COLOR 7:PRINT "<< マウス端子使用 簡易比色計の温度較正 プログラム >>"
2800 PRINT
2810 PRINT "*****"
2820 PRINT"この較正処理プログラムは計測する指示薬の比色を測定し、光センサーの定数を求"
2830 PRINT"めます。最初に測定しようとする濃度範囲の強酸側標準液を設定して計測します。"
2840 PRINT" 次に、中性標準液 (pH 7) に設定して計測し、較正データを決定します。"
2850 PRINT "*****"
2860 COLOR 6:LOCATE 4,10
2870 PRINT " 1.簡易比色計をコンピュータのマウス端子に接続してください。"
2880 COLOR 6:LOCATE 4,12
2890 PRINT " 確認したら、リターンキーを押して下さい。Nキーで中止します。";
2900 A$=INKEY$:IF A$="" THEN 2900
2910     IF A$="N" OR A$="n" OR A$="ミ" THEN GOTO *OWARI
2920 LOCATE 1,10:PRINT SPC(78);:LOCATE 1,12:PRINT SPC(78);
2930 COLOR 5
2940 LOCATE 4,10:PRINT " 2.測定予定範囲の強酸側の標準液(最小pH:5.2程度)を準備して下
    さい。"
2950 LOCATE 7,12:PRINT "準備した溶液のpHを入力して下さい。=>";
2960 INPUT C1
2970 WW=5:GOSUB *START:GOSUB *KEISAN1
2980 N1=COUNT
2990 '
3000 LOCATE 10,23:PRINT "* リターンキー で次の処理に進みます。";
3010 A$=INKEY$:IF A$="" THEN 3010
3020 '
3030 FOR I=10 TO 23:LOCATE 1,I:PRINT SPC(78);:NEXT I
3040 COLOR 4
3050 LOCATE 4,10:PRINT " 3.測定予定範囲の中性標準液 (pH:6.8程度)を準備して下さい。"
3060 LOCATE 7,12:PRINT "準備した溶液のpHを入力して下さい。=>";
3070 INPUT C2
3080 WW=5:GOSUB *START:GOSUB *KEISAN1
3090 N2=COUNT
3100 '
3110 FOR I=10 TO 23:LOCATE 1,I:PRINT SPC(78);:NEXT I
3120 *KEISAN2
3130 A=LOG(N1)-(LOG(N2)-LOG(N1))*C1/(C2-C1):B=(LOG(N2)-LOG(N1))/(C2
    -C1)
3140 COLOR 7:II=20

```

```

3150 LOCATE II,10:PRINT "p H =>";C1;"での測定カウント=>";N1
3160 LOCATE II,11:PRINT "p H =>";C2;"での測定カウント=>";N2
3170 LOCATE II,13:PRINT "測定したA定数(検量線切片) =>";A;
3180 LOCATE II,14:PRINT "          B定数(検量線傾き) =>";B;
3190 '
3200 LINE(30,256)-STEP(580,0),7,,&H9999
3210 COLOR 5:LOCATE 6,18:PRINT "<処理選択> ":COLOR 7
3220 LOCATE II,18:PRINT " 1 : dataを較正ファイルに保存し終了する"
3230 LOCATE II,20:PRINT " 2 : 再度、較正処理を行う"
3240 LOCATE II,22:PRINT " 3 : dataを保存せず終了する";
3250 A$=INKEY$:IF A$="" THEN 3250
3260     IF A$="1" THEN GOTO *FILESAVE
3270     IF A$="2" THEN GOTO 2770
3280     IF A$="3" THEN GOTO *SHORI ELSE 3250
3290 '
3300 '
3310 *START
3320 COLOR 6:LOCATE 60,12:PRINT "処理を実行中":COLOR 7
3330 FOR I=1 TO WW
3340   TM1=VAL(RIGHT$(TIME$,2))
3350   GOSUB *MOUSEIN
3360   TM2=VAL(RIGHT$(TIME$,2))
3370   IF TM2=TM1 THEN GOTO 3350
3380   LOCATE 20,I+12:PRINT "計測した COUNT(";I;") = ";COU
3390   C(I)=COU
3400 NEXT I
3410 RETURN
3420 '
3430 *MOUSEIN
3440 ' KT=0
3450 IF INP(&H7FD9) AND &H80 THEN 3470 ELSE 3450
3460 '
3470 COU=0
3480 *S1:IF INP(&H7FD9) AND &H80 THEN GOTO *S1 ELSE GOTO *S2
3490 *S2:COU=COU+1
3500 *S3:IF INP(&H7FD9) AND &H80 THEN GOTO *S4 ELSE GOTO *S2
3510 *S4:' KT=KT+1:IF KT=1 THEN 4170
3520 RETURN
3530 '
3540 '

```

```
3550 *FILESAVE
3560 COLOR 6:LOCATE 24,22:PRINT " *ただ今、ファイルに記録中です。*";
3570 OPEN FI$ FOR OUTPUT AS #1
3580   WRITE #1,A,B,C1,N1,C2,N2
3590 CLOSE #1
3600 GOTO *SHORI
3610 '
3620 *KEISAN1
3630 MA=0:MI=20:CC=0:CT=0
3640 FOR I=1 TO WW
3650   IF C(I)<MI THEN MI=C(I):XMI=I
3660   IF C(I)>MA THEN MA=C(I):XMA=I
3670 NEXT I
3680 FOR I=1 TO WW
3690   IF I=XMA OR I=XMI THEN 3710
3700   CC=CC+C(I)/100:CT=CT+1
3710 NEXT I
3720 COUNT=CC/CT:COUNT=100*COUNT
3730 RETURN
```

